NATURAL ENVIRONMENT RESEARCH COUNCIL

INSTITUTE of HYDROLOGY

REPORT No 13 DECEMBER 1971

A COMPUTER PROGRAM TO USE ORTHOGONAL POLYNOMIALS IN TWO VARIABLES FOR SURFACE-FITTING

by

Dorothy Richards

INSTITUTE OF HYDROLOGY
HOWBERY PARK
WALLINGFORD
BERKSHIRE

CONTENTS

LIST OF FIGURES

INSTITUTE OF HYDROLOGY

Report No. 13 December 1971

A COMPUTER PROGRAM TO USE ORTHOGONAL POLYNOMIALS IN TWO VARIABLES FOR SURFACE-FITTING

by Dorothy Richards

ABSTRACT

The program is written to run on any I.C.L. 1900 series computer that has a Fortran compiler. It will attempt to fit surfaces up to degree six, depending on the amount of data provided.

1. INTRODUCTION

Given the observed value of a function at n points (x,y) the program will fit the highest order polynomial surface up to degree six which is possible. The restriction is that the degrees of freedom of the residual term in the analysis of variance table is at least ten.

The program outputs the degree of fit attempted, the analysis of variance of the fit and the coefficients C in the resulting fitted surfaces which take the form

$$s = C_{10} + C_{11}x + C_{12}y$$

$$s = C_{20} + C_{21}x + C_{22}y + C_{23}x^2 + C_{24}xy + C_{25}y^2$$

.... up to the sextic surface, and the residuals.

The cost of running this program in the state shown in Fig.1 on the I.C.L. 1903A computer installed at the Hydraulics Research Station (HRS) for the example shown in Fig.2 was £1, the minimum charge.

2. INPUT TO THE PROGRAM

For users who are familiar with programming and the I.C.L. George System, there is sufficient detail on Fig.1 to run the program without reading this section further.

The input to the computer for this program is in two sections: first, the data-file; second, the computer job steering cards, henceforth referred to as job cards.

The data-file, as its name implies, holds the data which must be available to the program when it is executed under the control of the job cards. Hence the data-file precedes the job cards.

2.1 The data-file

The cards which the program expects to read, and the order in which it expects to read them, are listed below.

Card 1	The number of sets of data
2	A title card to the first set of data
3	The number N of observations in this set This must be not greater than one hundred (i.e. $N \leq 100$)
4 +	A card for each point giving grid coordinates (x,y) and the related observed value
4 + N	The title card to the second set of data
4 + N + 1	The number of observations in the second set
4 + N + 2 +,	Points and observations for set two
..... etc.,	

In this example there are two sets of data, the first containing 20 points (X,Y) together with their observed values Z; the second set has 25 points



FIG 1 An Example of the input cards

To ease data preparation, there is no fixed format on the numbers. There are three simple points to remember:-

(a) integers do not have decimal points,
(b) real numbers do have decimal points,
(c) one or more spaces, or a new card, will separate the data.

The first and last cards in this file, shown in Fig.1 are job cards and will be subject to restrictions noted in the following paragraph.

2.2 The job cards

The example of job cards in Fig.1 is relevant only to those intending to use the program on the I.C.L. 1903A under our present arrangement with H.R.S.; other installations will have their own operating procedures.

The items in capital letters are system words and should be left unchanged, although they may be abbreviated by the experienced user. The items in lower-case letters should be replaced by names that will help the operators and the user to identify the files and job. The system requires these identifiers to have no more than twelve characters and conventionally they begin with the user's initials.

The file BINOPS contains a compiled version of the program and is not the same as the listing given in Section 5 which is in source language.

Erasure of the data-file and the output-file is optional, but it is expensive to leave them in the computer file-store for any length of time, and the presence of many unused files slows computer throughput.

3. THE OUTPUT FROM THE COMPUTER

Fig. 2.1 is a copy of the data used to produce the output shown in Figs. 2.2 and 2.3.

Output is largely self-explanatory, but two symbols may be unfamiliar:

(i) **, exponentiation, or 'raised to the power of'.

(ii) the symbol rEm means that the number r is to be multiplied by 10 to the power m.

e.g. - 0.4055E 1 is - 4.055

0.4055E-1 is 0.04055

```
#LISTING OF :APROG.DRTESTDATA(1/)    PRODUCED ON 24FEB72 AT 19.48.03
#OUTPUT BY LISTFILE IN ':APROG.DRDR' ON 24FEB72 AT 19.54.31
DOCUMENT    DRTESTDATA

2
 TEST FIT Z=1-X-Y
20
 .9945    .5295   -.5240
 .9460    .4849   -.4309
 .4759    .8965   -.3724
 .8242    .7051   -.5293
 .6626    .2471    .0903
 .8139    .0641    .1220
 .9348    .3183   -.2531
 .1403    .4888    .3709
 .2104    .0996    .6900
 .6007    .0671    .3322
 .3222    .6053    .0725
 .3525    .2428    .4347
 .3394    .4223    .2383
 .1985    .0812    .7203
 .7693    .0879    .1428
 .0011    .2707    .7282
 .3142    .6471    .0377
 .4655    .4612    .0733
 .7518    .8871   -.6389
 .9122    .2901   -.2023
TEST FIT Z=(1-X*X-Y*Y)
 20
 .5295    .4973    .4722
 .4849    .4730    .5413
 .8965    .2379    .1397
 .3526    .4121   .7059
 .2471    .3313    .8290
 .0641    .4069    .8303
 .3183    .4674    .6803
 .4888    .0701    .7562
 .0996    .1052    .9789
 .0671    .3007    .9052
 .6053    .1611    .6076
 .2428    .1762    .9099
 .4223    .1697    .7928
 .0812    .0992    .9837
 .2707    .0005    .9268
 .0879    .3846    .8444
 .6471    .1571    .5567
 .4612    .2327    .7332
.8871    .3756    .0720
 .2901  .4561  .7078
****
```

Fig. 2. 1
Input data filed used to produce output in Figs. 2.2 & 2.3

ORTHOGONAL POLYNOMIAL FITTING
TEST FIT Z=1-X-Y
DEGREE OF FIT ATTEMPTED 3
ANALYSIS OF VARIANCE TABLE

SOURCE	DF	SS	MS	MSR
LINEAR FIT	2	3.40505	1.70253	33603.50395
ADDITIONS FOR				
QUADRATIC FIT	3	0.00010	0.00003	0.63816
CUBIC FIT	4	0.00022	0.00006	1.09696
RESIDUAL	10	0.00051	0.00005	
TOTAL	19	3.40588		

FITTED SURFACES

CONS X**3 X**2 Y**2 X**2 Y**3 X**3 Y**3	X X**2 Y X Y**3 X Y**4 X**2 Y**4	Y X Y**2 Y**4 Y**5 X Y**5	X**2 Y**3 X**5 X**6 Y**6	X Y X**4 X**4 Y X**5 Y	Y**2 X**3 Y X**3 Y**2 X**4 Y**2
0.1004E 01	-0.1003E 01	-0.1003E 01			
0.1000E 01	-0.9774E 00	-0.9973E 00	-0.3281E-01	0.2728E-01	-0.2557E-01
0.9663E 00 0.1510E 00	-0.8285E 00 0.1220E 00	-0.7805E 00 0.1613E 00	-0.2945E 00 0.2612E 00	-0.2448E 00	-0.4561E 00

Fig. 2.2
An example of output demonstrating a good linear fit

	RESIDUALS WHEN M=		
OBS	1	2	3
-0.5240	-0.0003	-0.0010	0.0013
-0.4309	-0.0000	0.0002	-0.0016
-0.3724	-0.0000	-0.0028	-0.0020
-0.5293	-0.0004	0.0017	-0.0059
0.0903	0.0016	0.0045	0.0046
0.1220	0.0018	-0.0015	-0.0025
-0.2531	0.0006	-0.0008	0.0011
0.3709	0.0023	-0.0000	0.0036
0.6900	0.0034	0.0043	0.0041
0.3322	0.0024	0.0033	0.0021
0.0725	0.0014	0.0018	-0.0013
0.4367	-0.0275	-0.0442	-0.0170
0.2383	0.0020	0.0048	0.0082
0.7203	0.0035	0.0041	0.0016
0.1428	0.0018	0.0002	-0.0006
0.7282	0.0035	-0.0007	-0.0024
0.0377	0.0023	0.0018	-0.0021
0.0733	0.0015	0.0055	0.0031
-0.6389	-0.0008	-0.0007	0.0045
-0.2023	0.0007	-0.0005	0.0011

Fig. 2.2 (Cont'd)

ORTHOGONAL POLYNOMIAL FITTING
TEST FIT Z=(1-X*X-Y*Y)
DEGREE OF FIT ATTEMPTED 3
ANALYSIS OF VARIANCE TABLE

SOURCE	DF	SS	MS	MSR
LINEAR FIT	2	1.09867	0.54933	45184144.10546
ADDITIONS FOR				
QUADRATIC FIT	3	0.07947	0.02649	2178857.43011
CUBIC FIT	4	0.00000	0.00000	0.72551
RESIDUAL	10	0.00000	0.00000	
TOTAL	19	1.17814		

FITTED SURFACES

CONS	X	Y	X**2	X Y	Y**2
X**3	X**2 Y	X Y**2	Y**3	X**4	X**3 Y
X**2 Y**2	X Y**3	Y**4	X**5	X**4 Y	X**3 Y**2
X**2 Y**3	X Y**4	Y**5	X**6	X**5 Y	X**4 Y**2
X**3 Y**3	X**2 Y**4	X Y**5	Y**6		

0.1179E 01	-0.8871E 00	-0.5288E 00			
0.1000E 01	-0.9060E-04	-0.6704E-03	-0.9997E 00	-0.5778E-03	-0.9984E 00
0.1001E 01	-0.3235E-02	-0.4945E-02	-0.9932E 00	0.6177E-02	-0.9822E 00
-0.4053E-02	-0.4348E-02	-0.4590E-02	-0.2036E-01		

Fig. 2.3
Further example of output demonstrating a quadratic fit

RESIDUALS WHEN M=

OBS	1	2	3
0.4722	-0.0258	0.0001	0.0001
0.5413	-0.0425	-0.0001	-0.0001
0.1397	0.1183	0.0000	0.0000
0.7059	-0.0575	-0.0001	-0.0000
0.8290	-0.0443	0.0002	0.0002
0.8303	0.0768	0.0001	0.0001
0.6803	-0.0307	-0.0001	-0.0001
0.7562	-0.0478	0.0000	-0.0000
0.9789	0.0562	0.0001	0.0001
0.9052	0.0554	-0.0001	-0.0001
0.6076	-0.0506	0.0001	0.0001
0.9099	-0.0393	0.0001	-0.0000
0.7928	-0.0780	0.0000	-0.0000
0.9837	0.0709	-0.0001	-0.0001
0.9268	0.0119	-0.0000	0.0000
0.8444	0.0534	-0.0000	-0.0000
0.5567	-0.0347	-0.0001	-0.0000
0.7332	-0.0863	-0.0001	-0.0000
0.0720	0.1215	-0.0000	-0.0000
0.7078	-0.0272	0.0000	-0.0000

Fig. 2.3 (Cont'd)

4. COMMENTS ON THE PROGRAM

This section is included for those who wish to understand the method and the program. There is no detailed flowchart but associated with each step in the method are the relevant line numbers of the program as listed in Section 5. Fig.3 is a sketch of the method used. The method for fitting orthogonal polynomials to the surface is described in Cadwell and Williams (1961).

Firstly, the terms involved in fitting a surface of degree six are written in the following pattern, numbering the rows m, each row containing m elements:-

$m = 1 : f_1(1)$
$m = 2 : f_2(x)\ f_3(y)$
$m = 3 : f_4(x^2)\ f_5(xy)\ f_6(y^2)$
$m = 4 : f_7(x^3)\ f_8(x^2y)\ f_9(xy^2)\ f_{10}(y^3)$
$m = 5 : f_{11}(x^4)\ f_{12}(x^3y)\ f_{13}(x^2y^2)\ f_{14}(xy^3)\ f_{15}(y^4)$
$m = 6 : f_{16}(x^5)\ f_{17}(x^4y)\ f_{18}(x^3y^2)\ f_{19}(x^2y^3)\ f_{20}(xy^4)\ f_{21}(y^5)$
$m = 7 : f_{22}(x^6)\ f_{23}(x^5y)\ f_{24}(x^4y^2)\ f_{25}(x^3y^3)\ f_{26}(x^2y^4)\ f_{27}(xy^5)\ f_{28}(y^6)$ (i)

The fundamental principles defining the orthogonality of these functions are

(i) $$\sum_{j=1}^{n} \{f_i(x_j, y_j)\}^2 = 1, \qquad i = 1, 2, 3 \ldots\ldots \text{(ii)}$$

(ii) $$\sum_{j=1}^{n} f_i(x_jy_j)\ f_k(x_jy_j) = 0, \qquad \left(\begin{array}{l} i \neq k \\ k = 1, 2, 3 \ldots\ldots \text{(iii)} \\ i = 1, 2, 3 \ldots\ldots \end{array} \right.$$

where n is the number of observations to be used in the fit. An iterative procedure for evaluating the functions of (i) is given in Forsythe (1957), and the iteration is covered by lines 42-86 of the program on listings 5.2 to 5.4.

Continuing to refer to the row number as m, and abbreviating $f_i(x,y)$ to f_i where the context is clear, then the iterative procedure is defined as

$$\lambda_j f_j = xf_{j-m+1} + e_{j1}f_1 + e_{j2}f_2 + \ldots. \; e_{j,j-1}f_{j-1} \quad \sum_{k=1}^{m-1} k < j < \sum_{k=1}^{m} k \qquad \text{(iv)}$$

and

$$\lambda_j f_j = yf_{j-m} + e_{j1}f_1 + e_{j2}f_2 + \ldots. \; e_{j,j-1}f_{j-1} \quad j = \sum_{k=1}^{m} k$$

To initialise the procedure

$$\sum_{i=1}^{n} f_1^{\,2} = 1, \quad \text{i.e.} \quad f_1 = \frac{1}{\sqrt{n}}$$

(Lines 29, 39, 40, listing 5.2)

To determine the e_{ji}, multiply the equations (iv) by f_i for $i = 1, 2 \ldots j-1$ and sum over n, which yields a set of equations of the form

$$\sum^{n} \lambda_j f_j f_i = \Sigma x f_{j-m+1} f_i + \sum_{t=1}^{j-1} e_{jt} \sum^{n} f_t f_i \qquad \text{(v)}$$

But $\Sigma f_j f_i = 0$ and $\Sigma f_t f_i = 0$ except when $t = i$, then $\Sigma f_t f_i = 1$. Thus from (v)

$$e_{ji} = - \sum_{k=1}^{n} x_k f_{j-m+1}(x_k, y_k)\; f_i(x_k y_k)$$

Note that the vector of values $x_k f_{j-m+1}(x_k, y_k)$ is involved in calculating f_j, as well as all the e_j of the j^{th} iteration, so it is useful to evaluate this vector once and hold it during the iteration. (Lines 46-55, listing 5.3)
The e_{ji} are determined in lines 56-58, listing 5.3.
Substituting for e_{ji} in (iv) the vector of elements $\lambda_j f_j$ is obtained. To satisfy equation (ii) for f_j

$$\sum^{n} (f_j)^2 = 1$$

$$\Sigma(\lambda_j f_j)^2 = \lambda_j^{\,2} \quad \Sigma f_j^{\,2} = \lambda_j^{\,2}$$

$$\therefore \lambda_j = \sqrt{\Sigma(\lambda_j f_j)^2}$$

which means we can now evaluate f_j. (Lines 64-69, listing 5.3)



Fig. 3

Block diagram of program

At this stage the coefficients of the terms in each individual orthogonal function f_i can be calculated.

$$f_1(1) = \frac{1}{\sqrt{n}} = w_{11} \text{ say}$$

$$f_2 = \frac{xf_1}{\lambda_2} + \frac{e_{21}f_1}{\lambda_2} = \frac{xw_{11}}{\lambda_2} + \frac{e_{21}w_{11}}{\lambda_2} = w_{21} + w_{22}x$$

$$f_j = \frac{xf}{\lambda_j}\ j\text{-}m\text{+}1 + w_{j1} + w_{j2}x + w_{j3}y + w_{j4}x^2 + \ldots\ldots.., \quad \text{(vi)}$$

Repetitive substitution shows

$$w_{ji} = \frac{1}{\lambda j} \sum_{i=1}^{j-1} \{ \sum_{k=i}^{j-1} e_{ji}\ w_{ki} \}$$

(Lines 70-73, listing 5.3)

But $\frac{x}{\lambda j}\ f_{j-m+1}$ also involves terms in 1, x, y, etc.,

Let j-m+1 = s, $\quad f_s = w_{s1} + w_{s2}x + w_{s3}y\ w_{s4}x^2 \quad \ldots\ldots..,$

$$\frac{x}{\lambda j}\ f_s = \frac{w_{s1}x}{\lambda_j} + \frac{w_{s2}x^2}{\lambda_j} + \frac{w_{s3}yx}{\lambda_j} + \ldots\ldots..,$$

Obviously the terms of f_j involving x (whilst excluding those in 1 and f(y), f(y^2) etc.) are incremented then by the terms of f_s divided by λ_j.

Likewise if equation (vi) was of the form $f_j = \frac{y}{\lambda_j}\ f_{j-m} + w_{j1} + w_{j2}x + w_{j3}y + \ldots,$

the terms of $f_s = f_{j-m}$ would have been adjusted into f_j, excluding the constant term and those involving x only.

(Lines 74-85, listing 5.3)

The terms $w_{j2} + \frac{w_{s1}}{\lambda_j}$ will now be referred to as v_{ji}.

At this stage the iterative procedure is finished and when performed sufficient times to give the required fit we have two matrices, one of the f elements and another of their coefficients, v.

Consider the equation

$$z = a_1f_1 + a_2f_2 + a_3f_3 + \ldots\ldots,$$

Successive multiplication by f_i, i = 1,2,3 ...., and summation over n, and with the application of equations (ii) and (iii) as previously used

$$\sum_{k=1}^{n} z_k f_i(x_k y_k) = a_i \qquad i = 1, 2, 3 \ldots\ldots,$$

and z_k are the observed values.

(Lines 87-91 listing 5.4)

There now remain three subroutines:

(a) SURFACES (Listing 5.7)
for a linear surface

$$z = a_1f_1 + a_2f_2 + a_3f_3$$

$$= a_1v_{11} + a_2(v_{21} + v_{22}x) + a_3(v_{31} + v_{32}x + v_{33}y)$$

$$= (a_1v_{11} + a_2v_{21} + a_3v_{31}) + (a_2v_{22} + a_3v_{32})x + a_3v_{33}y$$

To tidy up the surfaces, the elements of the function coefficients matrix v_{ij} are multiplied by their corresponding a_i, then the terms necessary for each degree of fit are summed. The remainder of the subroutine is printing.

(b) AOV (listing 5.8, 5.9)

This is a straightforward subroutine that does the necessary arithmetic to enable the following analysis of variance table to be output.

Analysis of variance of fit.

Source	d.f.	S.S.
Linear fit	2	$a_2^2 + a_3^2$
Additional for Quadratic	3	$a_4^2 + a_5^2 + a_6^2$
Cubic	4	$a_7^2 + a_8^2 + a_9^2 + a_{10}^2$
Quartic	5	$a_{11}^2 + a_{12}^2 + a_{13}^2 + a_{14}^2 + a_{15}^2$
Quintic	6	$a_{16}^2 + a_{17}^2 + a_{18}^2 + a_{19}^2 + a_{20}^2 + a_{21}^2$
Sextic	7	$a_{22}^2 + a_{23}^2 + a_{24}^2 + a_{25}^2 + a_{26}^2 + a_{27}^2 + a_{28}^2$
Residuals	By Subtraction	
Total	n-1	$\sum_{i=1}^{n} a_i^2 - a_1^2$

(c) RESIDUALS (listings 5.5, 5.6)

This subroutine points out the difference between the calculated and observed value at each of the input data points.

5. LISTING OF PROGRAM

```
FORTRAN COMPILATION BY #XFAT MK 4C    DATE  24/02/72  TIME  19/51/26

                 LIST(LP)
                 PROGRAM(H426 017)
                 INPUT 2=CR0
                 OUTPUT 1=LP0
                 COMPRESS INTEGER AND LOGICAL
                 END
```

```
      MASTER ORTHOPOLYS
      DIMENSION X(100),Y(100),PHI(100,28),F(28),BASIC(100),ALPHA(28),
     1  Z(100),TITLE(20),C(28,28)
      READ (2,102) NTIMES
      DO 30 IN=1,NTIMES
      READ (2,103) TITLE
      WRITE (1,104) TITLE
      READ (2,102) N
C     N IS NUMBER OF ELEMENTS,M DEGREE OF FIT,NTIMES NUMBER OF FITS
C     BOUNDS N 100 M 6
      IF (N-100) 17, 17,18
   18 WRITE (1,101)
      GO TO 30
   17 M=6
   21 NPHI=(M+1)*(M+2)/2
C     THE FOLLOWING PASSAGE ENSURES D.F. RESIDUALS IS AT LEAST 10
      IF (N-(NPHI+10)) 22,23,23
   22 M=M-1
      GO TO 21
   23 WRITE (1,105) M
      NPHI=NPHI-1
      READ (2,100) ((X(I),Y(I),Z(I)) I=1,N)
      A=1.0/SQRT(FLOAT(N))
      DO 37 I=1,100
      DO 38 J=1,28
   38 PHI(I,J)=0.0
   37 CONTINUE
      DO 35 I=1,28
      DO 36 J=1,28
   36 C(I,J)=0.0
   35 CONTINUE
      C(1,1)=A
      DO 1 I=1,N
    1 PHI(I,1)=A
      NI=2
      DO 2 NJ=1,NPHI
      NK=NJ+1
      DO 12 I=1,NJ
   12 F(I)=0.0
```

```
      NCOUNT=((NI+1)*NI)/2-1
      IM=NK+1-NI
      IF (NJ-NCOUNT) 3,4,3
    3 DO 5 I=1,N
    5 BASIC(I)=X(I)*PHI(I,IM)
      GO TO 6
    4 NI=NI+1
      IM=IM-1
      DO 7 I=1,N
    7 BASIC(I)=Y(I)*PHI(I,IM)
    6 DO 8 I=1,NJ
      DO 9 K=1,N
    9 E(I)=E(I)-BASIC(K)*PHI(K,I)
    8 CONTINUE
      DO 10 I=1,N
      DO 11 K=1,NJ
   11 PHI(I,NK)=PHI(I,NK)+E(K)*PHI(I,K)
   10 PHI(I,NK)=PHI(I,NK)+BASIC(I)
      RLAMBDA=0.0
      DO 13 I=1,N
   13 RLAMBDA=RLAMBDA+PHI(I,NK)*PHI(I,NK)
      RLAMBDA=1.0/SQRT(RLAMBDA)
   39 DO 14 I=1,N
   14 PHI(I,NK)=PHI(I,NK)*RLAMBDA
      DO 31 IC=1,NJ
```

```
      DO 32 INT=IC,NJ
   32 C(NK,IC)=C(NK,IC)+E(INT)*C(INT,IC)
   31 CONTINUE
      ID=2
      IF (NJ.EQ.NCOUNT) ID=ID+1
      MTOT=1
      DO 33 IS=1,IM
      C(NK,ID)=C(NK,ID)+C(IM,IS)
      MCOUNT=(MTOT*(MTOT+1))/2
      IF (IS-MCOUNT)33,0,33
      ID=ID+1
      MTOT=MTOT+1
   33 ID=ID+1
      DO 34 INT=1,NK
   34 C(NK,INT)=C(NK,INT)*RLAMBDA
    2 CONTINUE
      DO 15 I=1,NPHI+1
      ALPHA(I)=0.0
      DO 16 K=1,N
   16 ALPHA(I)=Z(K)*PHI(K,I)+ALPHA(I)
   15 CONTINUE
      CALL AOV(N,Z,ALPHA,NPHI,M)
      CALL SURFACES(C, ALPHA,NPHI+1,M)

      CALL RESIDUALS(X,Y,Z,C,M,N)
   30 CONTINUE
      STOP
  100 FORMAT (3F0.0)
  101 FORMAT (1X,32HELEMENTS BOUNDED IN ARRAY OF 100)
  102 FORMAT (I0)
  103 FORMAT (20A4)
  104 FORMAT (30H1ORTHOGONAL POLYNOMIAL FITTING/1X,20A4)
  105 FORMAT(1X,23HDEGREE OF FIT ATTEMPTED,I4)
      END

END OF SEGMENT, LENGTH   572, NAME  ORTHOPOLYS
```

```
      SUBROUTINE RESIDUALS(X,Y,OBS,C,M,N)
      DIMENSION X(100),Y(100),OBS(100),C(28,28),CALC(6)
      WRITE(1,100) (I,I=1,M)
  100 FORMAT(1H1,15X,'RESIDUALS WHEN M='/5X,'OBS',3X,6(5X,I1,4X))
      DO 99 I=1,N

      DO 30 J=1,M
   30 CALC(J)=0.0
      X1=X(I)
      Y1=Y(I)
      IF (M-1) 0,10,0
      X2=X1*X1
      XY=X1*Y1
      Y2=Y1*Y1
      IF (M-2) 0,10,0
      X3=X2*X1
      X2Y=X2*Y1
      XY2=X1*Y2
      Y3=Y2*Y1
      IF (M-3) 0,10,0
      X4=X2*X2
      X3Y=X3*Y1
      X2Y2=X2*Y2
      XY3=X1*Y3
      Y4=Y2*Y2
      IF (M-4) 0,10,0
      X5=X3*X2
      X4Y=X4*Y1
      X3Y2=X3*Y2
      X2Y3=X2*Y3
      XY4=X1*Y4
      Y5=Y3*Y2
      IF (M-5) 0,10,0
      X6=X3*X3
```

```
      X5Y=X5*Y1
      X4Y2=X4*Y2
      X3Y3=X3*Y3
      X2Y4=X2*Y4
      XY5=X1*Y5
      Y6=Y3*Y3
   10 CONTINUE

      CALC(1)=C(3,1)+C(3,2)*X1+C(3,3)*Y1
      IF (M-1) 0,20,0
      CALC(2)=C(6,1)+C(6,2)*X1+C(6,3)*Y1+C(6,4)*X2+C(6,5)*XY+C(6,6)*Y2
      IF (M-2) 0,20,0
      CALC(3)=
     *     C(10,1)+C(10,2)*X1+C(10,3)*Y1+C(10,4)*X2+C(10,5)*XY+C(10,6)*
     1Y2+C(10,7)*X3+C(10,8)*X2Y+C(10,9)*XY2+C(10,10)*Y3
      IF (M-3) 0,20,0
      CALC(4)=
     *     C(15,1)+C(15,2)*X1+C(15,3)*Y1+C(15,4)*X2+C(15,5)*XY+C(15,6)*
     1Y2+C(15,7)*X3+C(15,8)*X2Y+C(15,9)*XY2+C(15,10)*Y3+C(15,11)*X4+
     2C(15,12)*X3Y+C(15,13)*X2Y2+C(15,14)*XY3+C(15,15)*Y4
      IF (M-4) 0,20,0
      CALC(5)=
     *     C(21,1)+C(21,2)*X1+C(21,3)*Y1+C(21,4)*X2+C(21,5)*XY+C(21,6)*
     1Y2+C(21,7)*X3+C(21,8)*X2Y+C(21,9)*XY2+C(21,10)*Y3+C(21,11)*X4+
     2C(21,12)*X3Y+C(21,13)*X2Y2+C(21,14)*XY3+C(21,15)*Y4+C(21,16)*X5+
     3C(21,17)*X4Y+C(21,18)*X3Y2+C(21,19)*X2Y3+C(21,20)*XY4+C(21,21)*Y5
      IF (M-5) 0,20,0
      CALC(6)=
     *     C(28,1)+C(28,2)*X1+C(28,3)*Y1+C(28,4)*X2+C(28,5)*XY+C(28,6)*
     1Y2+C(28,7)*X3+C(28,8)*X2Y+C(28,9)*XY2+C(28,10)*Y3+C(28,11)*X4+
     2C(28,12)*X3Y+C(28,13)*X2Y2+C(28,14)*XY3+C(28,15)*Y4+C(28,16)*X5+
     3C(28,17)*X4Y+C(28,18)*X3Y2+C(28,19)*X2Y3+C(28,20)*XY4+C(28,21)*Y5+
     4C(28,22)*X6+C(28,23)*X5Y+C(28,24)*X4Y2+C(28,25)*X3Y3+C(28,26)*X2Y4
     5+C(28,27)*XY5+C(28,28)*Y6
   20 CONTINUE
      DO 97 K=1,M
   97 CALC(K)=CALC(K)-OBS(I)
   98 WRITE(1,101) (OBS(I),(CALC(K),K=1,M))
  101 FORMAT(1X,F10.4,6F10.4)
   99 CONTINUE
      RETURN
      END

END OF SEGMENT, LENGTH   718, NAME  RESIDUALS
```

```
      SUBROUTINE SURFACES(C,A,N,M)
      DIMENSION C(28,28), A(28)
      DO 1 I=1,N
      DO 2 J=1,I
    2 C(I,J)=C(I,J)*A(I)
    1 CONTINUE
      JJ=1
      DO 3 I=1,M
      J=((I+2)*(I+1))/2
      DO 4 IA=JJ,J-1
      DO 5 IB=1,J
    5 C(J,IB)=C(J,IB)+C(IA,IB)
    4 CONTINUE
    3 JJ=J
      WRITE (1,300)
      DO 6 I=1,M
      J=((I+1)*(I+2))/2
    6 WRITE (1,301) (C(J,K),K=1,J)
  300 FORMAT(//1X,15HFITTED SURFACES//18X,4HCONS,13X,1HX,14X,1HY,
     112X,4HX**2,11X,3HX Y,12X,4HY**2/18X,4HX**3,9X,6HX**2 Y,9X,
     26HX Y**2,11X,4HY**3,10X,4HX**4,10X,6HX**3 Y/15X,9HX**2 Y**2,8X,
     36HX Y**3,10X,4HY**4,11X,4HX**5,8X,7HX**4  Y,9X,9HX**3 Y**2/15X,
     49HX**2 Y**3,8X,6HX Y**4,10X,4HY**5,11X,4HX**6,8X,7HX**5  Y,9X,
     59HX**4 Y**2/15X,9HX**3 Y**3,7X,9HX**2 Y**4,6X,6HX Y**5,11X,
     64HY**6//)
  301 FORMAT(//(10X,6F15.4))
      RETURN
      END

END OF SEGMENT, LENGTH   159, NAME  SURFACES
```

```
      SUBROUTINE AOV(N,Z,A,NPHI,M)
      DIMENSION Z(N),A(28),RMS(7),RMSR(6)
      IDFR=N-NPHI-1
      IDFT=N-1
      SST=0.0
      DO 1 I=1,N
    1 SST=SST+Z(I)*Z(I)
      SST=SST-A(1)*A(1)
      SS1=A(2)*A(2)+A(3)*A(3)
      SSR=SST-SS1
      RMS(1)=SS1/2.0
      IF (M-1) 2,2,0
      SS2=A(4)*A(4)+A(5)*A(5)+A(6)*A(6)
      SSR=SSR-SS2
      RMS(2)=SS2/3.0
      IF (M-2) 2,2,0
      SS3=A(7)*A(7)+A(8)*A(8)+A(9)*A(9)+A(10)*A(10)
      SSR=SSR-SS3
      RMS(3)=SS3/4.0
      IF (M-3) 2,2,0
      SS4=A(11)*A(11)+A(12)*A(12)+A(13)*A(13)+A(14)*A(14)+A(15)*A(15)
      SSR=SSR-SS4
      RMS(4)=SS4/5.0
      IF (M-4) 2,2,0
      SS5=A(16)*A(16)+A(17)*A(17)+A(18)*A(18)+A(19)*A(19)+A(20)*A(20)+
     1    A(21)*A(21)
      SSR=SSR-SS5
      RMS(5)=SS5/6.0
      IF (M-5) 2,2,0
      SS6=A(22)*A(22)+A(23)*A(23)+A(24)*A(24)+A(25)*A(25)+A(26)*A(26)+
     1    A(27)*A(27)+A(28)*A(28)
      SSR=SSR-SS6
      RMS(6)=SS6/7.0
    2 RMS(7)=SSR/FLOAT(IDFR)
      DO 5 I=1,6
    5 RMSR(I)=RMS(I)/RMS(7)
```

```
      WRITE (1,200)
      WRITE (1,201) SS1,RMS(1),RMSR(1)
      IF (M-1) 4,4,0
      WRITE (1,202) SS2,RMS(2),RMSR(2)
      IF (M-2) 4,4,0
      WRITE (1,203) SS3,RMS(3),RMSR(3)
      IF (M-3) 4,4,0
      WRITE (1,204) SS4,RMS(4),RMSR(4)
      IF (M-4) 4,4,0
      WRITE (1,205) SS5,RMS(5),RMSR(5)
      IF (M-5) 4,4,0
      WRITE (1,206) SS6,RMS(6),RMSR(6)
    4 WRITE (1,207) IDFR,SSR,RMS(7)
      WRITE (1,208) IDFT,SST
  200 FORMAT(27H ANALYSIS OF VARIANCE TABLE//7H SOURCE,12X,2HDF,9X,2HSS,
     1 13X,2HMS,10X,3HMSR//)
  201 FORMAT(11H LINEAR FIT,9X,1H2,3F15.5/14H ADDITIONS FOR)
  202 FORMAT(14H QUADRATIC FIT,6X,1H3,3F15.5)
  203 FORMAT(10H CUBIC FIT,10X,1H4,3F15.5)
  204 FORMAT(12H QUARTIC FIT,8X,1H5,3F15.5)
  205 FORMAT(12H QUINTIC FIT,8X,1H6,3F15.5)
  206 FORMAT(11H SEXTIC FIT,9X,1H7,3F15.5/)
  207 FORMAT(9H RESIDUAL,9X,I3,2F15.5/)
  208 FORMAT(6H TOTAL,12X,I3,F15.5)
      RETURN
      END

END OF SEGMENT, LENGTH   495, NAME  AOV
```

REFERENCES

Cadwell, J.H. and Williams, D.E. 1961. Some orthogonal methods of curve and surface fitting. *The Computer Journal*, *4*, 260-264.

Forsythe, G.E. 1957. Generation and use of orthogonal polynomials for data fitting with a digital computer. *J. Soc. for Indust. and Appl. Maths.*, *5*, p.74.

REPORTS OF INSTITUTE OF HYDROLOGY

INSTITUTE OF HYDROLOGY

Report No. 1	1970	"Publications Handbook. Guide to preparation of manuscripts for publication. Papers for submission to external journals. Institute of Hydrology Reports", by B.C. Kennedy

HYDROLOGICAL RESEARCH UNIT

Report No. 2*	1967	"Snow in Great Britain", by P.C. Rowe
Report No. 3*	1968	"Evaporation from forests", by J.B. Stewart
Report No. 4*	1968	"Design of steep stream gauging structures for the Plynlimon experimental catchments", by P.W. Herbertson

INSTITUTE OF HYDROLOGY

Report No. 5	1969	"Processing errors in the analysis of catchment data", by A.N. Mandeville, D.T. Plinston and Miss A. Hill

* Out of print

Report No. 6	1969	"A feasibility study and development programme for continuous dilution gauging", by P.H. Hosegood, M.K. Bridle and P.W. Herbertson
Report No. 7	1969	"Installation of access tubes and calibration of neutron moisture meters", by C.W.O. Eeles
Report No. 8	(in press)	"The soil hydrology of the Plynlimon experimental catchments", by J.P. Bell
Report No. 9	1971	"River level sampling periods", by P.W. Herbertson, J.R. Douglas and Miss A. Hill
Report No. 10	1971	"User's testing schedule for the Wallingford Probe system", by P.M. Holdsworth
Report No. 11	1971	"Report on precipitation", by J.C. Rodda